ラプラス変換による微分方程式の解法

本ワークシートでの内容

- ラプラス変換コマンドの使用例を紹介します。
- 微分方程式を解くコマンドを紹介します。

- Mapleでラプラス変換を用いた微分方程式を解く方法と Maple の基本コマンドで微分方程式を解く例を紹介します。
- ラプラス変換などの積分変換には inttransパッケージを利用します。
- 微分方程式を解くには dsolve コマンドを使用します。
- 本ワークシートの構成
  - ラプラス変換による微分方程式解法
  - 組込み関数による解法

はじめに

主な利用コマンド

• **diff(式, 変数, 変換後の変数)**	微分
• **diff(式, 変数$n)**	n階微分
• **D(変数)(値)**	初期値の定義における微分
• **inttrans[laplace](式, 変数, 変換後の変数)**	ラプラス変換
• **map(作用子, 要素)**	要素の一括操作 使用例：**map(sin, [1, 2, 3])** = $[\sin(1), \sin(2), \sin(3)]$
• **solve(式, 変数)**	1つ以上の式を解く
• **dsolve({微分方程式, 初期値}, numeric, methhod=方法)**	常微分方程式(ODE)を解く
• **evalb(式)**	論理式の評価

ラプラス変換による微分方程式解法

次のような（連立）常微分方程式の解法を考えてみます。

微分方程式の定義

```
> dsys := diff(x(t), t$2) = -5*x(t)+2*y(t), diff(y(t), t$2) = 2*x(t)-2*y(t);
```

$$dsys := \ddot{x}(t) = -5\,x(t) + 2\,y(t), \ddot{y}(t) = 2\,x(t) - 2\,y(t) \tag{2.1}$$

初期値の定義

```
> init := x(0) = 0, D(x)(0) = 1, y(0) = 1, D(y)(0) = 0;
```

$$init := x(0) = 0, D(x)(0) = 1, y(0) = 1, D(y)(0) = 0 \tag{2.2}$$

Maple でラプラス変換を行うには、あらかじめ inttrans パッケージを読み込んでおく必要があります。

```
> with(inttrans);
```

$$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable] \tag{2.3}$$

このパッケージには他に各種のフーリエ変換などの関数も含められています。

さて、まずは与えられた微分方程式系 dsys をラプラス変換してみましょう。map 関数を用いて、各方程式に対してラプラス変換を適用しています。

```
> laplace_dsys := map(expr->laplace(expr, t, s), {dsys});
```

$$laplace\_dsys := \{s^2\, laplace(x(t), t, s) - D(x)(0) - s\,x(0) = -5\, laplace(x(t), t, s) + 2\, laplace(y(t), t, s), s^2\, laplace(y(t), t, s) - D(y)(0) - s\,y(0) = 2\, laplace(x(t), t, s) - 2\, laplace(y(t), t, s)\} \tag{2.4}$$

変換された系に対して、さらに初期条件を与えます。

```
> new_dsys := eval(laplace_dsys, {init});
```

$$new\_dsys := \{s^2\, laplace(x(t), t, s) - 1 = -5\, laplace(x(t), t, s) + 2\, laplace(y(t), t, s), s^2\, laplace(y(t), t, s) - s = 2\, laplace(x(t), t, s) - 2\, laplace(y(t), t, s)\} \tag{2.5}$$

変換された最終的な系 new_dsys において、各 laplace(x(t), t, s) および laplace(y(t), t, s) に関して方程式を記号的に解いてみます。

```
> sol := solve( new_dsys, {laplace(x(t), t, s), laplace(y(t), t, s)});
```

$$sol := \left\{laplace(x(t), t, s) = \frac{s^2 + 2s + 2}{s^4 + 7s^2 + 6}, laplace(y(t), t, s) = \frac{s^3 + 5s + 2}{s^4 + 7s^2 + 6}\right\} \tag{2.6}$$

各変換式に対して、s および t に関する有理式表現が得られました。最後に、これを逆ラプラス変換して元の関数に戻してみます。

```
> dsol1 := map(expr->invlaplace(expr, s, t), sol);
```

$$dsol1 := \left\{x(t) = \frac{2\cos(t)}{5} + \frac{\sin(t)}{5} - \frac{2\cos(\sqrt{6}\, t)}{5} + \frac{2\sqrt{6}\sin(\sqrt{6}\, t)}{15}, y(t) = \frac{4\cos(t)}{5} + \frac{2\sin(t)}{5} + \frac{\cos(\sqrt{6}\, t)}{5} - \frac{\sqrt{6}\sin(\sqrt{6}\, t)}{15}\right\} \tag{2.7}$$

これで微分方程式の解関数が得られました。

組込み関数による解法

前節では手動でラプラス変換を行って微分方程式の解法を試みましたが、実は Maple の微分方程式用の関数である dsolve には、ラプラス変換を用いて解関数を求めるための明示的なオプションが用意されています。

微分方程式の定義

```
> dsys;
```

$$\ddot{x}(t) = -5x(t) + 2y(t), \ddot{y}(t) = 2x(t) - 2y(t) \tag{3.1}$$

初期値の定義

```
> init;
```

$$x(0) = 0, D(x)(0) = 1, y(0) = 1, D(y)(0) = 0 \tag{3.2}$$

これら同じ系に対して、ラプラス変換を用いて解を求めるためには、dsolve 関数の method オプションを指定します。

```
> dsol2 := dsolve( {dsys, init}, {x(t), y(t)}, method=laplace);
```

$$dsol2 := \left\{x(t) = \frac{2\cos(t)}{5} + \frac{\sin(t)}{5} - \frac{2\cos(\sqrt{6}\, t)}{5} + \frac{2\sqrt{6}\sin(\sqrt{6}\, t)}{15}, y(t) = \frac{4\cos(t)}{5} + \frac{2\sin(t)}{5} + \frac{\cos(\sqrt{6}\, t)}{5} - \frac{\sqrt{6}\sin(\sqrt{6}\, t)}{15}\right\} \tag{3.3}$$

得られた解が、手動で行ったときのものと同じことは明らかです。

```
> map(rhs, dsol1)=map(rhs, dsol2);
```

$$\left\{\frac{2\cos(t)}{5}+\frac{\sin(t)}{5}-\frac{2\cos\left(\sqrt{6}\,t\right)}{5}+\frac{2\sqrt{6}\,\sin\left(\sqrt{6}\,t\right)}{15}, \frac{4\cos(t)}{5}+\frac{2\sin(t)}{5}+\frac{\cos\left(\sqrt{6}\,t\right)}{5}-\frac{\sqrt{6}\,\sin\left(\sqrt{6}\,t\right)}{15}\right\}$$
$$=\left\{\frac{2\cos(t)}{5}+\frac{\sin(t)}{5}-\frac{2\cos\left(\sqrt{6}\,t\right)}{5}+\frac{2\sqrt{6}\,\sin\left(\sqrt{6}\,t\right)}{15}, \frac{4\cos(t)}{5}+\frac{2\sin(t)}{5}+\frac{\cos\left(\sqrt{6}\,t\right)}{5}-\frac{\sqrt{6}\,\sin\left(\sqrt{6}\,t\right)}{15}\right\} \quad \textbf{(3.4)}$$

```
> evalb(%);
```

$$\mathit{true} \quad \textbf{(3.5)}$$